Find the isomorphisms between the finite fields
F = $\mathbb{Z}3[y]/(y^2+1)$ and G = $\mathbb{Z}3[y]/(y^2+2\cdot y+2)$ with 9 elements.
I'm using functions for the polynomials.

```
> p := 3;
  f := x -> x^2+1;
  g := x -> x^2+2*x+2;
```

$$p := 3$$
$$f := x \to x^2+1$$
$$g := x \to x^2+2\,x+2 \qquad (1)$$

Because now I can write

```
> f(y);
  g(z);
```

$$y^2+1$$
$$z^2+2\,z+2 \qquad (2)$$

By construction, [y] in F is a root of $f = x^2+1$. The other root is [2y].
We need to find the roots of $f(x)$ in G. So let's try them all.
Make sure you write g(y) here, to get the polynomial in y!

```
> F := [seq( seq( a*y+b, b=0..2 ), a=0..2 )];
  for a in F do
      a, Rem( f(a), g(y), y ) mod p;
  od;
```

$$F := [0, 1, 2, y, y+1, y+2, 2\,y, 2\,y+1, 2\,y+2]$$
$$0, 1$$
$$1, 2$$
$$2, 2$$
$$y, y+2$$
$$y+1, 0$$
$$y+2, 2\,y$$
$$2\,y, y+2$$
$$2\,y+1, 2\,y$$
$$2\,y+2, 0 \qquad (3)$$

There are two roots, $y+1$ and $2\cdot y+2$ hence two isomorphisms.

```
> beta1 := y+1; beta2 := 2*y+2;
```

$$\beta 1 := y+1$$
$$\beta 2 := 2\,y+2 \qquad (4)$$

Define $\varphi : F \to G$. You can do it this way in Maple using := which is nice.

```
> for a from 0 to 2 do
      for b from 0 to 2 do
          phi( a*y+b ) := a*beta1+b mod p;
      od
  od;
```

The following functionality of seq if nice. You iterate over all elements $a \in F$ by doing

```
> seq( a, a=F );
```

$0, 1, 2, y, y+1, y+2, 2\,y, 2\,y+1, 2\,y+2$ (5)

Now check that $\varphi(a+b) = \varphi(a) + \varphi(b)$ and $\varphi(a \cdot b) = \varphi(a) \cdot \varphi(b)$ for all a, b in F.
I'm going to use the function

```
> seq( seq( phi( Rem(a*b,f(y),y) mod 3 ) - Rem( phi(a)*phi(b), g(y),
  y ) mod 3, a=F ), b=F );
```

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, (6)
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0

```
> seq( seq( phi(a+b mod 3) - (phi(a) + phi(b) mod 3), a=F ), b=F );
```

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, (7)
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0

Exercise: Find the isomorphisms between F = $\mathbb{Z}2[y]/(y^3+y+1)$ and
G = $\mathbb{Z}2[y]/(y^3+y^2+1)$, finite fields with 8 elements.
Verify that $f(x) = x^3 + x^2 + 1$ has 3 roots in F and 3 roots in G.

```
>
>
```